Tail Calls

Announcements

Interpreting Scheme

The Structure of an Interpreter

Eval

Base cases:

- Primitive values (numbers)
- Look up values bound to symbols

Recursive calls:

- Eval(operator) & Eval(operand) of calls
- Apply(procedure, arguments)
- Eval(sub-expression) of special forms

Requires an environment for symbol lookup

Apply

Base cases:

- Built-in primitive procedures

Recursive calls:

- Eval(body) of user-defined procedures

Creates a new environment each time a user-defined procedure is applied

Project 4

Pairs in Project 4: Scheme

https://cs61a.org/proj/scheme/ (released on Wed.)

Tokenization/Parsing: Converts text into Python representation of Scheme expressions:

- Numbers are represented as numbers
- Symbols are represented as strings
- Lists are represented as instances of the Pair class

Evaluation: Converts Scheme expressions to values while executing side effects:

- scheme_eval(expr, env) returns the value of an expression in an environment
- scheme_apply(procedure, args) applies a procedure to its arguments
- The Python function scheme_apply returns the return value of the procedure it applies

(Demo)

Dynamic Scope

Dynamic Scope

The way in which names are looked up in Scheme and Python is called lexical scope (or static scope) [You can see what names are in scope by inspecting the definition]

Lexical scope: The parent of a frame is the environment in which a procedure was *defined*

Dynamic scope: The parent of a frame is the environment in which a procedure was *called*

Special form to create dynamically scoped procedures (you will implement **mu** special form in Project 4 Scheme)

mu

(define f (~~lambda~~ (x) (+ x y)))

(define g (lambda (x y) (f (+ x x))))

(g 3 7)



Lexical scope: The parent for f's frame is the global frame

Error: unknown identifier: y

Dynamic scope: The parent for f's frame is g's frame

13

Space Efficiency

Space and Environments

Which environment frames do we need to keep during evaluation?

At any moment there is a set of active environments

Values and frames in active environments consume memory

Memory that is used for other values and frames can be recycled

Active environments:

- Environments for any function calls currently being evaluated
- Parent environments of functions named in active environments

Fibonacci Space Consumption



Fibonacci Space Consumption

Has an active environment
Can be reclaimed
Hasn't yet been created



fib takes **linear space.**

(Demo)

pythontutor.com/composingprograms.html#code=def%20fib%28n%29%3A%0A%20%20%20%20if%20n%20%3D%3D%200%20or%20n%20%3D%3D%201%3A%0A%20%20%20%20%20%20%20%20return%20n%0A%20%20%20%20else%3A%0A%20%20%20%20%20%20%20%20return%20fib%28n-2%29%20%2B%20fib%28n-1%29%0A%20%20%20%20%20%20%20%20%0Afib%286%29&mode=display&origin=composingprograms.js&cumulative=false&py=3&rawInputLstJSON=[]&curInstr=1

Tail Recursion

Functional Programming

All functions are pure functions.

No re-assignment and no mutable data types.

Name-value bindings are permanent.

Advantages of functional programming:

- The value of an expression is independent of the order in which sub-expressions are evaluated
- Sub-expressions can safely be evaluated in parallel or only on demand (lazily) (Demo)
- **Referential transparency:** The value of an expression does not change when we substitute one of its subexpression with the value of that subexpression

But... no for/while statements! Can we make recursion efficient? Yes!

Recursion and Iteration in Python

In Python, recursive calls always create new active frames.

fact_k(n, k) computes: n! * k

```
def fact_k(n, k):
    if n == 0:
        return k
    else:
        return fact_k(n - 1, n*k)
```

```
def fact_k(n, k):
    while n > 0:
        n, k = n - 1, k * n
    return k
```

Time	Space
Linear	Linear
Linear	Constant

Tail Recursion

From the Revised[7] Report on the Algorithmic Language Scheme:

> "Implementations of Scheme are required to be properly tail-recursive. This allows the execution of an iterative computation in constant space, even if the iterative computation is described by a syntactically recursive procedure."

```scheme
(define (fact_k n k)
  (if (= n 0) k
    (fact_k (- n 1)
            (* k n))))
```

Should use resources like

```python
def fact_k(n, k):
    while n > 0:
        n, k = n-1, k*n
    return k
```

How? Eliminate the middleman!

Time	Space
Linear	Constant

(Demo)

pythontutor.com/composingprograms.html#code=def%20factorial%28n,%20k%29%3A%0A%20%20%20%20if%20n%20%3D%3D%200%3A%0A%20%20%20%20%20%20%20%20return%20k%0A%20%20%20%20else%3A%0A%20%20%20%20%20%20%20%20return%20factorial%28n-1,%20k*n%29%0A%20%20%20%20%0Afactorial%284,%2010%29&mode=display&origin=composingprograms.js&cumulative=true&py=3&rawInputLstJSON=[]&curInstr=0

Tail Calls

Tail Calls, Tail Contexts, Tail Recursion

A procedure call that has not yet returned is active. Some procedure calls are tail calls. A Scheme interpreter should support an unbounded number of active tail calls using only a constant amount of space.

A tail call is a call expression in a ***tail context***:

- The last body sub-expression in a **lambda** expression (or procedure definition)
- Sub-expressions 2 & 3 in a tail context **if** expression
- All non-predicate sub-expressions in a tail context **cond**
- The last sub-expression in a tail context **and**, **or**, **begin**, or **let**

A recursive procedure is tail recursive if ***all*** of its recursive calls are tail calls

```
(define (fact-k n k)
  (if (= n 0) k
    (fact-k (- n 1)
            (* k n)) ) )
```

```
(define fact-k (lambda (n k)
  (if (= n 0) k
    (fact-k (- n 1)
            (* k n)) ) ))
```

Example: Length of a List

```
(define (length s)
  (if (null? s) 0
    (+ 1 (length (cdr s))) ) )
```

Not a tail context

A call expression is not a tail call if more computation is still required in the calling procedure

Linear recursive procedures can often be re-written to use tail calls

```
(define (length-tail s)
  (define (length-iter s n)
    (if (null? s) n
      (length-iter (cdr s) (+ 1 n)) ) )
  (length-iter s 0) )
```

Recursive call is a tail call

Break: 5 minutes

Tail Recursion Examples

Which Procedures are Tail Recursive?

Which of the following procedures run in constant space?

```scheme
;; Compute the length of s.
(define (length s)
  (+ 1 (if (null? s)
           -1
           (length (cdr s))) ) )
```

```scheme
;; Return whether s has any repeated elements.
(define (has-repeat2 s)
  (if (null? s)
      #f
      (if (contains (cdr s) (car s))
          #t
          (if (has-repeat2 (cdr s))
            #t
            #f                          )    )   ) )
```

```scheme
;; Return whether s contains v.
(define (contains s v)
  (if (null? s)
      #f
      (if (= v (car s))
          #t
          (contains (cdr s) v))))
```

```scheme
;; Return whether s has any repeated elements.
(define (has-repeat s)
  (if (null? s)
      #f
      (if (contains (cdr s) (car s))
          #t
          (has-repeat (cdr s))) ) )
```

Tail Recursion Practice: sum-digits

(Demo)

Tail Recursion with Scheme Lists

```
(define (map procedure s)
  (if (null? s)
      nil
      (cons (procedure (car s))
            (map procedure (cdr s))) ) )
```

```
(map (lambda (x) (- 5 x)) (list 1 2))
```




```
(define (map procedure s)
  (define (map-reverse s m)
    (if (null? s)
        m
        (map-reverse (cdr s)
                     (cons (procedure (car s))
                           m))              ) )
  (reverse (map-reverse s nil)))
```

```
(define (reverse s)
  (define (reverse-iter s r)
    (if (null? s)
        r
        (reverse-iter (cdr s)
                      (cons (car s) r)) ) )
  (reverse-iter s nil))
```

Tail Recursion Techniques

Base case should return the complete answer (rather than a partial solution).

Define a helper with an extra parameter to keep track of progress so far.

Sketch an iterative solution (e.g. in Python) — names that are iteratively updated need to be tracked as function arguments in recursion.

Verify all recursive calls are tail calls.

(Demo)